
1
Introduction

During the RAN5#83 TTCN sidebar and in several TTCN CRs, SS Vendors have reported that in some EN-DC test cases TTCN requests DL user-plane data transmission on DRB of the NR cell while the UE under test may have not yet completed the Random Access procedure on this NR cell. The issue has also been  reported when TTCN reconfigure SS to stop responding to RACH preambles.
This issue happens after an RRCConnectionReconfiguration message is transmitted on E-UTRA side and data is sent to UE on the NR side, that can happen in the test case preamble of when there is a reconfiguration in the test body. As per 3GPP TS 38.331 subclause 5.3.5.3 NOTE 1, for EN-DC the UE may send the RRCConnectionReconfigurationComplete message (embedding the RRCReconfigurationComplete message) on the E-UTRA serving cell before the Random Access procedure has been started/completed on the NR cell.  

Therefore the issue may or may not occur depending on the UE speed in performing the Random Access procedure on the NR cell, and also depending on the SS speed in executing the TTCN code. A general solution is needed, that is independent from those factors.
2
Proposal

It is proposed to use the existing RACH Preamble Indication ASP in the following sequence in the NR PTC:

· Reconfigure SS to enable NR RACH Preamble indication (to TTCN)

· Send NR RRC information ( CellGroupConfig, RadioBearerConfig, …) to the E-UTRA PTC to transmit the RRCConnectionReconfiguration message

· Wait in parallel the RRCReconfigurationComplete indication from E-UTRA PTC and the RACH preamble indication

· Reconfigure SS to disable NR RACH preamble indication

It is assumed that at the time RACH preamble indication is reported to TTCN the Random Access procedure is completed by SS and UE.

The changes will be documented in a TTCN CR.

2.1 
Example of TTCN implementation in RLC TC 7.1.2.2.1
Here is an example of how it will be implemented in TC 7.1.2.2.1, in the common function f_NR_SendRRCReconfigurationContentsToEUTRA:


- Add a parameter “boolean p_WaitForRachPreamble” 

- When the parameter is true, SS is reconfigured to report RACH preamble
     - Wait for the RACH preamble indication after having sent the NR RRC ie to E-UTRA PTC, 

- After its reception RACH preamble reporting is disabled in the SS.

function f_NR_SendRRCReconfigurationContentsToEUTRA(template (omit) RadioBearerConfig p_RadioBearerConfig,

                                                      template (omit) CellGroupConfig p_SCGConfig,

                                                      template (omit) MeasConfig p_MeasConfig := omit,

                                                      template (omit) RadioBearerConfig p_RadioBearerConfig2 := omit,

                                                      boolean p_WaitForRRCReconfigurationComplete := true,

                                                      boolean p_WaitForRachPreamble := false
                                                      ) runs on NR_BASE_PTC

  {

    //@sic R5-195240: added p_WaitForRRCReconfigurationComplete sic@

    var template (omit) octetstring v_EncodedSCG_Config := omit;

    var template (value) RRCReconfiguration v_Reconfiguration;

    var RRCReconfigurationComplete v_Complete;

    var octetstring v_EncodedRRCReconfig;

    var bitstring v_EncodedComplete;

    var octetstring v_EncodedRB_Config;

    var octetstring v_OctetData;

    var boolean v_SendReconfigurationMsg := false;

    var integer v_Result;

    var EUTRA_NR_Coordination_MSG v_ReceivedMsg;    
    //Enable RACH Preamble Indication

    if (p_WaitForRachPreamble) {

       f_NR_SS_SystemIndCtrlConfig(nr_Cell1, cds_NR_SystemIndCtrl_RachPreamble(enable));      

    }
    // RRC Reconfiguration only sent if either SCG Config or Meas Config included

    if (isvalue (p_SCGConfig) or isvalue (p_MeasConfig)) {

      v_SendReconfigurationMsg := true;

      if (isvalue (p_SCGConfig)) {

        v_EncodedSCG_Config := bit2oct(encvalue(p_SCGConfig));

      }

      v_Reconfiguration := cs_38508_RRCReconfigurationIE (tsc_NR_RRC_TI_Def, omit, v_EncodedSCG_Config, p_MeasConfig);

      v_EncodedRRCReconfig := bit2oct(encvalue(v_Reconfiguration));

      f_EUTRA_NR_SendCoOrd(EUTRA, cms_EUTRA_NR_OctetData (v_EncodedRRCReconfig, "Reconfig"));

    }

    if (isvalue (p_RadioBearerConfig)) { // @sic R5s190115 sic@

      v_EncodedRB_Config := bit2oct(encvalue(p_RadioBearerConfig));

      f_EUTRA_NR_SendCoOrd(EUTRA, cms_EUTRA_NR_OctetData (v_EncodedRB_Config, "RB"));

    }

    if (isvalue (p_RadioBearerConfig2)) {

      v_EncodedRB_Config := bit2oct(encvalue(p_RadioBearerConfig2));

      f_EUTRA_NR_SendCoOrd(EUTRA, cms_EUTRA_NR_OctetData (v_EncodedRB_Config, "RB2")); // @sic R5s190062 sic@

    }

    






    alt  {

        [v_SendReconfigurationMsg and p_WaitForRRCReconfigurationComplete] EUTRA.receive(cmr_EUTRA_NR_OctetData( "Complete")) -> value v_ReceivedMsg {

            // If we sent the reconfiguration message, then we need to check the complete sent back

            v_OctetData := v_ReceivedMsg.OctetData.Data;

            v_EncodedComplete := oct2bit(v_OctetData);

            v_Result := decvalue(v_EncodedComplete, v_Complete);

            if (v_Result != 0 or not match (v_Complete, cr_38508_RRCReconfigurationCompleteIE)) {

                FatalError(__FILE__, __LINE__, "Decoding error");

            }

            if (p_WaitForRachPreamble){

                SYSIND.receive(car_NR_PRACH_Preamble_IND(?, cr_TimingInfo_Any, ?))

                f_NR_SS_SystemIndCtrlConfig(nr_Cell1, cds_NR_SystemIndCtrl_RachPreamble(disable))

            }

        }

        [p_WaitForRachPreamble]SYSIND.receive(car_NR_PRACH_Preamble_IND(?, cr_TimingInfo_Any, ?)) {

            f_NR_SS_SystemIndCtrlConfig(nr_Cell1, cds_NR_SystemIndCtrl_RachPreamble(disable));   

            if (v_SendReconfigurationMsg and p_WaitForRRCReconfigurationComplete){

                EUTRA.receive(cmr_EUTRA_NR_OctetData( "Complete")) -> value v_ReceivedMsg;

                // If we sent the reconfiguration message, then we need to check the complete sent back

                v_OctetData := v_ReceivedMsg.OctetData.Data;

                v_EncodedComplete := oct2bit(v_OctetData);

                v_Result := decvalue(v_EncodedComplete, v_Complete);

                if (v_Result != 0 or not match (v_Complete, cr_38508_RRCReconfigurationCompleteIE)) {

                    FatalError(__FILE__, __LINE__, "Decoding error");

                }
            }

        }

    }
  }

Handle the new parameter p_WaitForRachPreamble in the wrapper common functions

  function f_NR_ENDC_ReConfigAM_UM(NR_CellId_Type p_CellId,

                                   integer p_NoOfNR_AMBearersN,

                                   integer p_NoOfNR_UMBearersM,

                                   template (omit) NR_RadioBearerList_Type p_SS_Drb_ConfigList := omit,

                                   template (omit) NR_RLC_BearerToAddModList_Type p_RLC_BearerConfigList := omit,

                                   template (omit) SRB_ToAddModList p_SRB_ToAddModList := omit,

                                   template (omit) DRB_ToAddModList p_DRB_ToAddModList := omit,

                                   template (omit) SecurityConfig p_SecurityConfig := omit,

                                   template (omit) MAC_CellGroupConfig p_MAC_CellGroupConfig := omit,

                                   boolean p_WaitForRachPreamble := false
                                   ) runs on NR_BASE_PTC

  {

    var DRB_Identity v_DrbId := tsc_NR_DRB2;    // for ENDC only

    var template (value) CellGroupConfig v_SCGConfig;

    var template (value) RadioBearerConfig v_RadioBearerConfig;

    var template (value) NR_RLC_BearerToAddModList_Type v_RLC_BearerConfigList;

    var template (value) DRB_ToAddModList v_DRB_ToAddModList;

    var template (value) SecurityConfig v_SecurityConfig;

    var template (value) NR_RadioBearerList_Type v_SS_Drb_ConfigList;

    var template (value) RLC_Config v_RLC_Config;

    var integer i;

    var boolean v_IsFR1:= f_NR_CellInfo_GetIsFR1(p_CellId);

    for (i:=0; i < p_NoOfNR_AMBearersN + p_NoOfNR_UMBearersM; i:=i+1) {

      if (i < p_NoOfNR_AMBearersN) {

        v_SS_Drb_ConfigList[i] := f_NR_GetRadioBearerConfig_FullAM(p_CellId, v_DrbId);

        v_RLC_Config := cs_38508_RLC_Config_AM_DRB(f_NR_UL_AM_RLC(v_IsFR1), f_NR_DL_AM_RLC(v_IsFR1));

      } else {

        v_SS_Drb_ConfigList[i] := f_NR_GetRadioBearerConfig_FullUM(p_CellId, v_DrbId); //ENDC function: Sdap=omit

        v_RLC_Config := cs_38508_RLC_Config_UM_DRB(f_NR_UL_UM_RLC(), f_NR_DL_UM_RLC(v_IsFR1));

      }

      v_RLC_BearerConfigList[i] := cs_38508_RLC_BearerConfig_DRB(v_DrbId, v_RLC_Config, cs_38508_LogicalChannelConfig_DRB);

      v_DRB_ToAddModList[i] := cs_38508_DRB_ToAddMod(v_DrbId, cs_38508_PDCP_Config);

      v_DrbId := v_DrbId + 1;

    }

    v_SecurityConfig := cs_38508_SecurityConfig(cs_38508_SecurityAlgorithmConfig (f_NR_AS_CipheringAlgorithm_Get(), f_NR_AS_IntegrityAlgorithm_Get()), secondary);

    if (isvalue(p_SS_Drb_ConfigList)) {

      v_SS_Drb_ConfigList := valueof(p_SS_Drb_ConfigList);

    }

    if (isvalue(p_RLC_BearerConfigList)) {

      v_RLC_BearerConfigList := valueof(p_RLC_BearerConfigList);

    }

    if (isvalue(p_DRB_ToAddModList)) {

      v_DRB_ToAddModList := valueof(p_DRB_ToAddModList);

    }

    if (isvalue(p_SecurityConfig)) {

      v_SecurityConfig := valueof(p_SecurityConfig);

    }

    f_NR_SS_CommonRadioBearerConfig(p_CellId, v_SS_Drb_ConfigList);

    v_RadioBearerConfig := cs_38508_RadioBearerConfigDef(p_SRB_ToAddModList, v_DRB_ToAddModList, -, v_SecurityConfig);

    v_SCGConfig := f_NR_GetSCGConfigDef(p_CellId);

    if (isvalue(p_MAC_CellGroupConfig)) {

      v_SCGConfig.mac_CellGroupConfig := p_MAC_CellGroupConfig;

      f_NR_SS_ServingCellConfig(p_CellId, cs_TimingInfo_Now, cs_NR_ServingCellConfig_SpCell(cs_NR_SS_SpCellConfig(omit, omit, p_MAC_CellGroupConfig, omit)));

    }

    v_SCGConfig.rlc_BearerToAddModList := v_RLC_BearerConfigList;

    f_NR_SendRRCReconfigurationContentsToEUTRA(v_RadioBearerConfig, v_SCGConfig, -, -, -, p_WaitForRachPreamble);

  }

  function f_NR_ENDC_ConfigAM_UM(NR_CellId_Type p_CellId,

                                 integer p_NoOfNR_AMBearersN,

                                 integer p_NoOfNR_UMBearersM,

                                 template (omit) NR_RadioBearerList_Type p_SS_Drb_ConfigList := omit,

                                 template (omit) NR_RLC_BearerToAddModList_Type p_RLC_BearerConfigList := omit,

                                 template (omit) SRB_ToAddModList p_SRB_ToAddModList := omit,

                                 template (omit) DRB_ToAddModList p_DRB_ToAddModList := omit,

                                 template (omit) SecurityConfig p_SecurityConfig := omit,

                                 template (omit) MAC_CellGroupConfig p_MAC_CellGroupConfig := omit,

                                 boolean p_WaitForRachPreamble := false
                                ) runs on ENDC_NR_PTC

  {

    // Get Security info from EUTRA, calculate latest keys and configure SS

    // as only need to do this for the first reconfiguration

    f_ENDC_GetKeysActivateSecurity(p_CellId, EUTRA);

    f_NR_ENDC_ReConfigAM_UM(p_CellId,

                            p_NoOfNR_AMBearersN,

                            p_NoOfNR_UMBearersM,

                            p_SS_Drb_ConfigList,

                            p_RLC_BearerConfigList,

                            p_SRB_ToAddModList,

                            p_DRB_ToAddModList,

                            p_SecurityConfig,

                            p_MAC_CellGroupConfig,

                            p_WaitForRachPreamble);

  }
Pass the added parameter as true in the preamble to wait for RACH preamble.
  function f_TC_7_1_2_2_1_ENDC_NR() runs on ENDC_NR_PTC

  { // UM RLC / Segmentation and reassembly / 6-bit SN / Segmentation Info (SI) field

    var NR_RLC_SS_State_Type v_RLC_Rec;

    var template (omit)  NR_RadioBearerList_Type v_SS_Drb_ConfigList;

    var template (value) NR_RLC_BearerToAddModList_Type v_RLC_BearerConfigList;

    var template (value) DRB_ToAddModList  v_DRB_ToAddModList;

    f_ENDC_NR_Init();

    //Create and configure NR cell1

    f_NR_CellConfig_Def(nr_Cell1);

    f_NR_ENDC_PreambleOnEUTRA(nr_Cell1, MCG_Only);

    //Set null cipher Algorithm

    f_NR_AS_CipheringAlgorithm_Set(nea0); // @sic R5s190254 ch. 1.1 sic@

    v_SS_Drb_ConfigList[0] := cs_NR_SS_RadioBearer_UM_Test(tsc_NR_DRB2, size6); // SNsize 6

    v_DRB_ToAddModList[0] := cs_38508_DRB_ToAddMod(tsc_NR_DRB2, cs_38508_PDCP_Config); // PDCP default is 18 bits from 38.508

    v_RLC_BearerConfigList := f_NR_ReconfigureRLC_UM (tsc_NR_DRB2, size6, size6, f_NR_DL_UM_T_Reassembly(f_NR_CellInfo_GetIsFR1(nr_Cell1)));

    // Get NR capabilities sent during EUTRA preamble

    f_NR_ENDC_ConfigAM_UM(nr_Cell1, 0, 1, v_SS_Drb_ConfigList, v_RLC_BearerConfigList, -, v_DRB_ToAddModList , -, -, true);

    // Initialization of RLC variables

    f_NR_InitRLC_Record(v_RLC_Rec);

    v_RLC_Rec.UM_SN_Size := 6;

    if (not f_NR_CellInfo_GetIsFR1(nr_Cell1)) {v_RLC_Rec.t_Reassembly := 0.03;};

    // Synchronize with the EUTRA PTC

    f_EUTRA_NR_WaitForCoOrd_Trigger(EUTRA, "Start TestBody"); // @sic R5s190163 sic@


    f_NR_TestBody_Set(true);

    f_TC_7_1_2_2_1_NR_TestBody (v_RLC_Rec, tsc_NR_DRB2);

    f_EUTRA_NR_SendCoOrd (EUTRA, cms_EUTRA_NR_Trigger);

    f_NR_TestBody_Set(false);

    //Wait from coordination AFTER the connection is released in E-UTRA

    f_EUTRA_NR_WaitForCoOrd_Trigger(EUTRA, "Postamble");

    //Postamble

    f_NR_ReleaseAllCells();

  }
2.2
List of test cases
Here is the list of test cases in which the issue had been reported and in which this solution will be applied. If more test cases are affected a TTCN CR shall be raised and the same solution shall be applied.
RLC test cases

· f_TC_7_1_2_2_1_ENDC_NR 
· f_TC_7_1_2_2_2_ENDC_NR
· f_TC_7_1_2_2_3_ENDC_NR
· f_TC_7_1_2_2_4_ENDC_NR
· f_TC_7_1_2_2_5_ENDC_NR
· f_TC_7_1_2_2_6_ENDC_NR
· f_TC_7_1_2_3_1_ENDC_NR
· f_TC_7_1_2_3_2_ENDC_NR
· f_TC_7_1_2_3_3_ENDC_NR
· f_TC_7_1_2_3_4_ENDC_NR
· f_TC_7_1_2_3_5_ENDC_NR
· f_TC_7_1_2_3_6_ENDC_NR
· f_TC_7_1_2_3_7_ENDC_NR
· f_TC_7_1_2_3_8_ENDC_NR
· f_TC_7_1_2_3_9_ENDC_NR
· f_TC_7_1_2_3_10_ENDC_NR
· f_TC_7_1_2_3_11_ENDC_NR
PDCP test case

· f_TC_7_1_3_1_1_ENDC_NR
· f_TC_7_1_3_1_2_ENDC_NR
· fl_ENDC_PDCP_Integrity
· fl_ENDC_PDCP_Ciphering
· f_TC_7_1_3_4_1_ENDC_NR
· f_TC_7_1_3_4_2_ENDC_NR

· fl_TC_7_1_3_4_2_NR_TestBody
· f_TC_7_1_3_5_1_ENDC_NR
· f_TC_7_1_3_5_2_ENDC_NR
· f_TC_7_1_3_5_3_ENDC_NR
· f_TC_7_1_3_5_4_ENDC_NR
RRC test cases

· fl_TC_8_2_5_1_1_ENDC_NR_TestBody
· fl_TC_8_2_5_4_1_ENDC_NR_TestBody

